w/om23

AA Eastern Region Computer Programs
d Problems NWS ERCP - No. 9

DEPARTMENT OF COMMERCE
UNITED STATES OF AMERICA

STABILITY ANALYSIS PROGRAM

Scientific Services Division
Eastern Region Headquarters
March 1983

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Weather Service

NOAA Technical Memorandum
National Weather Service, Eastern Region Computer Programs and Problems

The Eastern Region Computer Programs and Problems (ERCP) series is a subset of the Eastern Region Technical Memorandum series. It will serve as the vehicle for the transfer of information about fully documented AFOS application programs. The format of ERCP - No. 1 will serve as the model for future issuances in this series.

1 An AFOS version of the Flash Flood Checklist. Cynthia M. Scott, March 1981. (PB81 211252).

2 An AFOS Applications Program to Compute Three-Hourly Stream Stages. Alan P. Blackburn, September 1981. (PB82 156886).

3 PUPPY (AFOS Hydrologic Data Reporting Program). Daniel P. Provost, December 1981. (PB82 199720).

4 Special Search Computer Program. Alan P. Blackburn, April 1982. (PB83 175455).

5 Conversion of ALEMBIC$ Workbins. Alan P. Blackburn, October 1982. (PB83 138313).

6 Real-Time Quality Control of SAOs. John A. Billet, January 1983. (PB83 166082).

7 Automated Hourly Weather Collective from HRR Data Input. Lawrence Cedrone, January 1983. (PB83 167122).

8 Decoders for FRH, FTJ and FD Products. Cynthia M. Scott, February 1983. (PB83 176057).

NOAA Eastern Region Computer Programs and Problems - No. 9

STABILITY ANALYSIS PROGRAM

Hugh M. Stone

Scientific Services Division
Eastern Region Headquarters
March 1983

UNITED STATES DEPARTMENT OF COMMERCE
Malcolm Baldrige, Secretary
/
National Oceanic and Atmospheric Administration
John V. Byrne, Administrator
/
National Weather Service
Richard E. Hallgren,
Acting Assistant Administrator

STABILITY ANALYSIS PROGRAM

Hugh M. Stone
Scientific Services Division
National Weather Service Eastern Region
Garden City, New York

I. General Information

A. Summary

An attempt has been made to develop a new indicator of static stability which would be more highly correlated with convective activity than any of the commonly used stability indices, such as lifted index, K index, total-total index, etc. All these indices are easy to compute, but all suffer from the same deficiency that only a few levels of the raob sounding are utilized. It is believed that a better indicator of stability should consider the complete vertical distribution of temperature and moisture. A simple computation is no longer important since a computer is now available.

The technique proposed here examines the relative importance of energy areas on a thermodynamic diagram. The method consists of selecting a parcel of air in lowest 150mb of the atmosphere which has maximum instability (maximum wet bulb potential temperature), then computing the energy gained or lost by that parcel as it is raised to the upper levels of the atmosphere, while entraining environmental air during its ascent. This process utilizes all information available on the vertical structure of the atmosphere, and should provide a good measure of stability.

The program was developed to run on an Eclipse S/230 minicomputer. Language used is Data General FORTRAN IV.

II. Application

A. Complete Program Description

This program produces a stability analysis of second transmission raob data (UJ1, significant levels), using the parcel method with entrainment, and computes a variety of stability indices.

A parcel of air is selected having maximum wet bulb potential temperature in the lowest 150mb of the sounding. This pressure level is denoted PMAX. This parcel is selected for several reasons. First, it is almost always the parcel that is most unstable with respect to large displacements in the atmosphere; i.e., it can make the ascent to the upper levels of the atmosphere with the minimum energy required for forced lift, or the maximum energy release, as the case may be. It is also frequently the most unstable with respect to small perturbations. The parcel is usually above the morning inversion on the 12Z raobs, and is typical of the parcels that may be involved in convective activity during the afternoon.

Applying Newton's second law to the parcel, and assuming hydrostatic equilibrium of the environmental air, it can be shown (Haltiner and Martin, 1957) that the change in kinetic energy per unit mass of a parcel moving from pressure p_1 to pressure p_2, is given by,

$$\Delta KE = - R_d \int_{p_1}^{p_2} (T_p - T_e) \, d(\ln p)$$

where R_d = gas constant for dry air

T_p = virtual temperature of parcel

T_e = " " " environment

In this program temperature is used rather than virtual temperature, since it is sufficiently accurate for the computation. The parcel is raised from level PMAX to the top of the sounding PTOP, usually 100mb. Entrainment is allowed at a rate which produces a doubling of mass in a 500mb ascent. Positive and negative energy areas are computed and integrated over various depths of the atmosphere.

The first index obtained, B1, represents the total energy gained or lost by the parcel as it is moved from the level PMAX to 400mb. The positive and negative areas are also integrated separately with results printed out as B1P and B1N, respectively. The B1 index is suitable for use as a stability index at anytime during the year.

A second index B2 is computed which represents the total energy change of a parcel moving from level PMAX to the equilibrium level, EL. EL is defined as the base pressure of the uppermost negative area computed during the ascent. During the convective season the EL is typically near or below the tropopause. During the winter the EL is usually much lower and may even coincide with level PMAX, when extremely stable conditions prevail and the parcel is in a negative area during its entire ascent. In such a case the B2 index is undefined, and the value 99999 is output to indicate this. If EL is above level PMAX, a value of B2 will be computed, but it will not be a good measure of stability, if EL is at a very low level below the tropopause, which is usually the case during the winter season. Therefore, the index B2 should be used only during the convective season, at which time it may represent stability better than index B1. The index B2 is composed of positive and negative parts which are output separately as B2P and B2N, respectively.

As a byproduct of the energy calculations described above, the lifting condensation level LCL of the parcel is obtained, also the level of free convection LFC is computed, if defined. The LFC is determined as the pressure at the top of the first negative area during the ascent. During very stable conditions, commonly during the winter season, the LFC may be undefined, in that case a value of zero is assigned.

Base and top pressure for each positive or negative area are determined and output as P1 and P2, respectively, along with the energy value for each area. The energy area which is intersected by the 400mb level has an energy value EX computed for that portion of the area below 400mb. This is used in the computation of the B1 index.

Lifted index LI, K index KI, Showalter index SWI, and convective condensation level CCL are all computed in the standard way (Sadowski and Rieck, 1977). These stability indices may differ from the NMC values--sometimes computed from mandatory level data (US1). The mean mixing ratio for the lowest 100mb of the atmosphere is computed and printed out as variable WAVG. The CCL is the pressure at which this mean mixing ratio line intersects the sounding temperature, excluding any possible intersections due to a low level inversion. The convective temperature is obtained by following the dry adiabat through this point to the surface pressure. The energy area between this dry adiabat and the sounding curve below the CCL is proportional to the solar heating required in the lower levels of the atmosphere to initiate convection. This energy area is denoted ETCCL on the output.

Wet bulb potential temperatures are also computed for all significant levels for the purpose of evaluating the potential (convective) stability of layers, which may become saturated, either through lifting or by precipitation falling through the layer. A potentially unstable layer is defined as one that has wet bulb potential temperature increasing with pressure (decreasing with height). All potentially unstable layers are identified, and the one having greatest depth is output as the last line of AFOS product CCCWRKTPA (Fig. 1) along with the lapse rate TWLAPSE (Deg. C/100mb) of wet bulb potential temperature through the layer.

A detailed listing of potentially unstable layers is given in AFOS product CCCWRKTPB (Fig. 2). Base and top pressure of layers are denoted by P1 and P2, respectively. The depth of the layer in millibars is given as DP. The lapse rate of wet bulb potential temperature is denoted TWLAPSE (Deg. C/100mb), and finally, the amount of lifting in millibars required to bring the base and top of the layer to saturation is given by DP1 and DP2, respectively. The unstable layers are followed by a complete listing of all significant levels.

The output from this program can be used in conjunction with the plotted raob to assess the static stability in the vicinity of the raob station. The determination of positive and negative energy areas provides a more consistent and quantitative evaluation of the raob than a subjective examination of the plotted sounding only.

The vertical total of the energy areas between PMAX and PX and between PMAX and EL are given by the indices B1 and B2, respectively. Limited experience with the index B1 during the winter season over the eastern United States suggests a good correspondence between large values of this index and convective activity. However, the atmospheric stability cannot be truly represented by a single number. The vertical distribution of energy areas may be more important than the sum of energies represented by B1 or B2; for example, a large negative area in the lower atmosphere may act to inhibit convection even though a larger positive area is present above. Positive and negative parts of the B1 and B2 indices are useful in assessing this latent instability.

The equilibrium level EL, not the tropopause level, is an important parameter to consider when severe weather is a possibility (Doswell, et al., 1982). Thunderstorm tops overshooting the EL are usually an indication of a severe storm.

The variable ETCCL, to be used with the 12Z sounding, is a measure of the amount of solar heating required for the surface temperature to reach the convective temperature, where theoretically convection can begin. Large values of ETCCL tend to delay or suppress convection completely.

The last line of the product CCCWRKTPA shows the deepest potential (convective) unstable layer and the lapse rate of wet bulb potential temperature through the layer. It is for use when large scale lifting of atmospheric layers is expected at any time of the year, not just the convective season. Details of the layers and amount of lift required to achieve saturation are given in product CCCWRKTPB.

B. Machine Requirements

This program requires 20K core in background. Total runtime usually varies from 25 to 45 seconds. Two FORTRAN channels are open during the program run. One RDOS channel is used at the end of the computation to insert headings and endings for the two AFOS products created. Disk space required for the program is 40448 bytes (79 RDOS blocks).

C. Database

Products that are referenced:

1. CCCSGLXXX : actual raob product.

Files/products that are created:

1. INDEXX : temporary storage file for index data computed which are eventually stored as AFOS products CCCWRKTPA. INDEXX is deleted at the end of the computation.

2. INDEXY : temporary storage file for potentially unstable layers and significant level data which are eventually stored as AFOS product CCCWRKTPB. INDEXY is also deleted at the end of the computation.

D. Structure of Software

RAN is the main program. It computes wet bulb potential temperature for all the significant levels and also makes the assumption that if a dewpoint is missing above 700mb, the atmosphere is dry at that point, and calculates a dewpoint 30°C less than the temperature. If a dewpoint is missing below 700mb, computation is terminated, and the alert light is turned on. All output is done by RAN. All other computations are done in the various subroutines. Interpolations and extrapolations of temperature and dewpoint in any of the subroutines are done assuming a linear variation of the quantity with the logarithm of pressure.

The function of various subroutines are as follows:

<u>DECOS</u>

Reads the temperature portion of the UJ1 raob specified in the array JST, utilizing the AFREAD subroutine (Peroutka, 1981).

TEMP1

This subroutine called by DECOS for decoding temperature and dewpoint.

INDX1

Computes lifted index, K index, and Showalter index.

BNDX

Determines pressure level PMAX that has highest wet bulb potential temperature in the lowest 150mb of the sounding. If an identical maximum value is found at 2 levels, the lowest level (highest pressure) is selected for PMAX. A modified raob is created which has its base at level PMAX and an additional significant level is added at level PX = 400mb, if a significant level does not already exist there. If the raob terminates below level PX, but within 50mb of PX, a level PX is extrapolated, so that the index B1 may still be computed.

RANN2

The principal subroutine, does all the energy area computations as the parcel is raised from the bottom to the top of the sounding. The first half of the subroutine raises a parcel along a dry adiabat from PMAX to the LCL level. Since entrainment is allowed during the dry ascent, the LCL is usually slightly higher than it would be if there were no entrainment.

The second half of the subroutine continues a moist ascent above the LCL. Entrainment is continued all the way using the method proposed by Austin (1948).

A 50mb step is used for the parcel ascent, but if a significant level is present within the next 50mb, the step is reduced to terminate at that level. Steps are likewise shortened, if the energy area changes sign, or if the LCL is reached during the next 50 millibars. The 50mb step was selected because it gives sufficient accuracy and is economical in computer time.

The stability indices B1 and B2 are computed after the ascent has been completed. Equilibrium level EL and level of free convection are also determined here.

The second call to subroutine RANN2 calculates the variable ETCCL, which represents the amount of low level heating required for a surface parcel to reach the convective temperature and then move dry adiabatically to the convective condensation level, CCL. No entrainment is allowed in the calculation of ETCCL.

CCL1

Computes the convective condensation level CCL, the mean mixing ratio for the lowest 100mb, and convective temperature, defined in Section A.

MODRB

Called immediately after CCL1 is finished to modify the original raob so ETCCL can be computed by a second call to RANN2. An additional significant level is added at the CCL level and the starting point of the parcel ascent is changed to the convective temperature and the surface pressure.

PULYR

Determines all potential (convective) unstable layers and computes lapse rate of wet bulb potential temperature in the layers and the amount of lifting required to achieve saturation at the base and top pressure of the layer.

WOBF SATLFT TCONOF WMROF DPTOF VAPFW

All thermodynamic computations are done using these six subroutines from the National Severe Storms Forecast Center, Kansas City, Mo. (Doswell, et al., 1982).

III. Procedures

A. Preparation

AFOS products CCCWRKTPA and CCCWRKTPB must be added to the database (several versions for each). The program RAN.SV must be moved to the main disk DPØ or DPØF with a link to DPØ. The program can operate on any significant level raob stored in the database.

B. Initiating the Program

To run the program for a single raob station, at the ADM console, type:

RUN:RAN CCCSGLXXX

where CCCSGLXXX is the actual nine-letter AFOS identifier desired.

If running the program for several stations, a simple procedure shown in Fig. 4 may save some time. If the procedure is used, change RNN to RUN and type the nine-letter AFOS identifier for the raob station. Wait for the alert light to come on, before striking the "X" which displays the results. The procedure cycles on itself.

C. Output

Output results are displayed on the ADM, if the procedure is used. A sample of the output with corresponding plotted raob is shown in Figures 1, 2, and 3, respectively.

D. Cautions or Restrictions on Use

If the raob data is not in proper format or raob terminates below 450mb, a message is printed on the DASHER, the alert light is usually turned on and computation is terminated. AFOS products CCCWRKTPA and CCCWRKTPB should contain several versions, so results of previous runs can be saved for a while.

E. Program Listings.

See Page 10.

ACKNOWLEDGMENT

Thanks to Cynthia Scott for programming advice, and Claudia Morales for the typing of this manuscript.

REFERENCES

Haltiner, G. J. and F. L. Martin, 1957: Dynamical and Physical Meteorology, McGraw-Hill Book Co., 470 pp. (see pp. 57-63).

Sadowski, A. F. and R. E. Rieck, 1977: Stability Indices, Technical Procedures Bulletin No. 207, National Weather Service, Silver Spring, Md.

Peroutka, M., 1981: Accessing the AFOS Database, NOAA Western Region Computer Programs and Problems NWS WRCP - No. 23, National Weather Service, Salt Lake City, Ut.

Austin, J. M., 1948: A Note on Cumulus Growth in a Nonsaturated Environment, Journal of Meteorology, 5, 103-107.

Doswell, C. A., J. T. Schaefer, D. W. McCann, T. W. Schlatter, and H. B. Wobus, 1982: Thermodynamic Analysis Procedures at the National Severe Storms Forecast Center, Preprints, 9th Conf. Weather Forecasting and Analysis, Amer. Meteor. Soc., Seattle, 304-309.

RAOB ANALYSIS FOR PBI DATE = 8 HOUR = 12 UNITS : J/KG X 10-1
EFF = 100.000 PERCENT OF ENTRAINMENT PER 500MB ASCENT

P0 = 1016. PTOP = 100. PMAX (MAX INSTABILITY) = 922. PX = 400.
EL = 412. LCL = 915. LFC = 807. BASED ON PARCEL MVG FM LVL -PMAX-

B2	=	20.	ENERGY PMAX TO EL	B1	=	18.	ENERGY PMAX TO PX
B2P	=	25.	POSITIVE PART	B1P	=	25.	POSITIVE PART
B2N	=	-5.	NEGATIVE PART	B1N	=	-7.	NEGATIVE PART

P1	P2	ENERGY GAINED (LOST) IN LAYER
922.	807.	-3.
807.	726.	2.
726.	657.	-2.
657.	412.	23.
412.	100.	-528.

EX = -2.

LI = -4. KI = 12. SWI = -5.

CCL = 842. ETCCL = 24. CONV TEMP = 29.3 (84.8F) WAVG = 11.66 G/KG

DEEPEST POT. UNSTABLE LYR : 787. - 708. MB, TWLAPSE = 7.4 SEE WRKTPB

Figure 1. Example of CCCWRKTPA.

* *

POTENTIAL (CONVECTIVE) UNSTABLE LAYERS FOR PBI DATE = 8 HOUR = 12

P1	P2	DP	TWLAPSE	DP1	DP2
500.	486.	14.	0.7	83.	192.
787.	708.	79.	7.4	5.	266.
922.	850.	72.	1.7	7.	57.
1016.	974.	42.	0.6	5.	70.

SIGNIFICANT LEVELS

P	T	TD	TW (WET BULB POTENTIAL TEMP)
100.	-67.3	-97.3	31.5
107.	-64.3	-94.3	31.3
116.	-66.7	-96.7	29.2
163.	-61.7	-91.7	24.2
300.	-38.9	-42.6	18.7
406.	-23.7	-24.3	17.6
417.	-26.7	-29.8	15.1
440.	-23.7	-35.7	14.3
486.	-17.3	-47.3	13.5
500.	-17.3	-28.3	13.6
708.	5.4	-24.6	12.1
726.	5.6	-11.4	13.1
758.	6.6	4.8	16.9
787.	8.6	8.2	18.0
850.	14.2	9.6	17.9
922.	16.2	15.7	19.1
956.	19.6	15.1	18.5
974.	19.2	14.2	17.1
1016.	18.2	17.9	17.4

Figure 2. Example of CCCWRKTPB.



Figure 3. Plotted raob showing various paramenters corresponding to Figures 1 and 2.

	DISPLAY (1-4)	MODE (D/M)	ACC/OV (R/A/O)	COMMAND (ANY COMMAND; LAST LINE MUST BE END OR "NAME")
01				RNN:RAN CCCSGLXXX
02				WCHR X
03	1	D	R	WRKTPA
04				WCHR X
05	1	D	R	WRKTPB
06				"032"
07				
08				
09				
10				

Figure 4. Procedure "032" for running "RAN" program.

```
C   RAN
C   RAOB STABILITY ANALYSIS
C   USING PARCEL METHOD, WITH SELECTED ENTRAINMENT RATE AND PRESSURE STEP
       COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
       COMMON/G/PP(0:20),ET(20),TW(0:50),DP,EFF,KMOD,KK
       COMMON/GG/NJ,PPB(15),PPT(15),DELPP(15),DTWDP(15),DPB(15),DPT(15),
     1  PTMAX,PBMAX,TWLAPSE,DMAX
       COMMON/T/RLCL,RLFC,EL,B2,B2P,B2N,IALL,B1,B1P,B1N,EX
       COMMON/CCL/PCCL,ETCCL,TS0,TSD0,L,TSCCL,TCCL,TDCCL,WAVG
       COMMON/TT/PT(0:50),TST(0:50),TSDT(0:50)
       COMMON/V/JNOM,PX
       COMMON/H/IHDR1(11),IHDR2(11)
       INTEGER DAT(32),SW(2)
       DATA IHDR1/"ERHWRKTPA000",177777K,177777K,"70",142600K,6412K/
       DATA IHDR2/"ERHWRKTPB000",177777K,177777K,"70",142600K,6412K/
       THETA(T,P2,P1)=T*(P2/P1)**.2857142 ; DRY ADIABATIC (T,P1) TO (THETA,P2)
       IOUT=6412K    ;   CR LF
       IEND=101603K  ;   ENDING FOR AFOS PRODUCT
       DP=50.        ;  50 MB STEP
       EFF=100.      ;  100% ENTRAINMENT PER 500MBS
       IALL=1
       CALL FOPEN (20,"INDEXX",300)
       CALL FOPEN (21,"INDEXY",300)
       CALL FCOM (IC,IER)
       CALL COMCM (IC,DAT,11,SW,IER)
       CALL COMCM (IC,DAT,11,SW,IER)
       CALL KLOSE (IC,IER)
       DO 3 I=1,5
3      JST(I)=DAT(I)
       PX=400.
       KMOD=1
       CALL DECOS($28)
       DO 76 I=0,JNO
       IF (TSD(I).NE.999.) GO TO 76
       IF (P(I).GE.700.) GO TO 20
       TSD(I)=TS(I)-30.  ;  IF DEWPT MISG ABV 700MB, ASSUME DRY
76     CONTINUE
       GO TO 6
20     WRITE (10,21) (JST(J),J=1,5),P(I)
21     FORMAT (1H ,5A2," DEWPOINT MISSING AT P = ",F5.0)
       CALL FORKE ("STAB","DEWPT",IER)  ;  TURN ON ALERT LIGHT
       GO TO 28
6      CALL INDX1 (RLI,RKI,RWI,$28)
       IF (IALL.EQ.0) GO TO 97
C
C   COMPUTE WET BULB POTENTIAL TEMP AT ALL SGFNT LEVELS ABV SFC
       DO 78 I=0,JNO
       TC=TCONOF(TS(I),TSD(I))  ;  CONDENSATION TEMPERATURE
       TH=THETA(TS(I)+273.16,1000.,P(I))-273.16  ;  POT TEMP DEG C
       WTH=WOBF(TH)
       WTC=WOBF(TC)
78     TW(I)=TH-WTH+WTC  ;  WET BULB POT TEMP - DEG C
C
       CALL PULYR  ;  DETERMINES POTENTIAL (CONVECTIVE) UNSTABLE LAYERS
       WRITE (21,1) (JST(I),I=4,5),JDATE,JHOUR
1      FORMAT (12X,"              POTENTIAL (CONVECTIVE) UNSTABLE LAYERS FOR "
     1  ,2A2,"    DATE = ",I3,"    HOUR = ",I3)
       WRITE (21,2)
2      FORMAT ("<15><12>",3X,"P1",8X,"P2",8X,"DP",4X,"TWLAPSE",6X,"DP1",7X,"DP2")
       M=NJ+1
```

```
      DO 8 I=1,NJ
      J=M-I
      WRITE (21,7) PPB(J),PPT(J),DELPP(J),DTWDP(J),DPB(J),DPT(J)
7     FORMAT ("<15><12>",1X,F5.0,2F10.0,F10.1,2F10.0)
8     CONTINUE
      WRITE (21,5)
      WRITE (21,77)
77    FORMAT ("<15><12>","SIGNIFICANT LEVELS",
     1 /"<15><12>",4X,"P",9X,"T",9X,"TD",8X,"TW (WET BULB POTENTIAL TEMP)")
      M=JNO
      DO 79 I=0,M
      J=M-I
      WRITE (21,80) P(J),TS(J),TSD(J),TW(J)
80    FORMAT ("<15><12>",1X,F5.0,3F10.1)
79    CONTINUE
      WRITE (21,5)  ;  ENDING WITH BLANK LINE
97    CALL BNDX ($28)  ;  ADJUST ORIGINAL RAOB FOR MAX INSTABILITY
      CALL RANN2 (PT,TST,TSDT,JNOM,PX)  ;  COMPUTE ENERGY INDICES
      WRITE (20,54) (JST(I),I=4,5),JDATE,JHOUR
54    FORMAT (12X,"          RAOB ANALYSIS FOR ",2A2,"   DATE = ",I3,"   HOUR = "
     1 ,I3,11X,"UNITS : J/KG X 10-1")
      WRITE (20,71) EFF
71    FORMAT ("<15><12>",5X,"EFF = ",F7.3,3X,"PERCENT OF ENTRAINMENT PER 500MB ASCENT")
      WRITE (20,5)
5     FORMAT ("<15><12>")  ;  PUTTING IN BLANK LINE
      WRITE (20,4) P(0),P(JNO),PT(0),PX
4     FORMAT ("<15><12>","P0 = ",F5.0,4X,"PTOP = ",F5.0,4X,"PMAX (MAX INSTABILITY) = ",
     1 F5.0,4X,"PX = ",F5.0)
      WRITE (20,50) EL,RLCL,RLFC
50    FORMAT ("<15><12>","EL = ",F5.0,5X,"LCL = ",F5.0,4X,"LFC = ",F5.0,4X,
     1 "BASED ON PARCEL MVG FM LVL -PMAX-")
      WRITE (20,5)
      WRITE (20,56) B2,B1,B2P,B1P,B2N,B1N
56    FORMAT ("<15><12>","B2  = ",F7.0,2X,"ENERGY PMAX TO EL",5X,"B1  = ",F7.0,2X,
     1 "ENERGY PMAX TO PX"/
     2 "<15><12>","B2P = ",F7.0,2X,"POSITIVE PART",9X,"B1P = ",F7.0,2X,"POSITIVE PART"/
     3 "<15><12>","B2N = ",F7.0,2X,"NEGATIVE PART",9X,"B1N = ",F7.0,2X,"NEGATIVE PART")
      WRITE (20,5)
      WRITE (20,59)
59    FORMAT ("<15><12>","P1",9X,"P2",9X,"ENERGY GAINED (LOST) IN LAYER")
      KK1=KK-1
      DO 53 I=1,KK1
      J=I-1
      WRITE (20,52) PP(J),PP(I),ET(I)
52    FORMAT ("<15><12>",F5.0,5X,F5.0,5X,F7.0)
53    CONTINUE
      WRITE (20,100) EX
100   FORMAT ("<15><12>","EX = ",F7.0)
      WRITE (20,5)
      WRITE (20,81) RLI,RKI,RWI
81    FORMAT ("<15><12>","LI = ",F4.0,4X,"KI = ",F4.0,4X,"SWI = ",F4.0)
      CALL CCL1
      CALL MODRB
      CALL RANN2 (P,TS,TSD,JNO,PX)  ;  COMPUTES ETCCL ONLY AND RETURNS
      WRITE (20,5)
      TS0F=1.8*TS0+32.  ;  CONV TEMP IN DEG F
      WRITE (20,93) PCCL,ETCCL,TS0,TS0F,WAVG
93    FORMAT ("<15><12>","CCL = ",F5.0,2X,"ETCCL = ",F6.0,2X,"CONV TEMP = "
     1 ,F5.1," (",F5.1,"F )  WAVG = ",F5.2," G/KG")
      WRITE (20,5)
```

```
      IF (DMAX.GT.0.) GO TO 9
      WRITE (20,10)
10    FORMAT ("<15><12>","DEEPEST POT. UNSTABLE LYR : NONE")
      GO TO 11
9     WRITE (20,12) PBMAX,PTMAX,TWLAPSE
12    FORMAT ("<15><12>","DEEPEST POT. UNSTABLE LYR : ",F5.0," - ",F5.0,
     1 " MB, TWLAPSE = ",F5.1," SEE WRKTPB")
      WRITE (20,5)  ;  ENDING WITH A BLANK LINE
11    CALL CLOSE (20,IER)
      CALL CLOSE (21,IER)
      CALL GCHN (ICHN,IER)  ;  GET RDOS CHANNEL
      CALL OPENN (ICHN,"INDEXX",0,IER)
      CALL WRS (ICHN,IHDR1,22,IER)  ;  HEADER INSERTION
      CALL KLOSE (ICHN,IER)
      CALL GCHN (ICHN,IER)  ;  GET RDOS CHANNEL
      CALL OPENA (ICHN,"INDEXX",0,IER)
      CALL WRS (ICHN,IEND,2,IER)  ;  ENDING FOR AFOS PRODUCT
      CALL KLOSE (ICHN,IER)
      CALL FSTORE ("INDEXX",0,IER)  ;  STORE INTO ERHWRKTPA
      CALL FORKP ("STAB","ERHWRKTPA",IER)  ;  TURN ON ALERT LIGHT
      CALL GCHN (ICHN,IER)  ;  GET RDOS CHANNEL
      CALL OPENN (ICHN,"INDEXY",0,IER)
      CALL WRS (ICHN,IHDR2,22,IER)  ;  HEADER INSERTION
      CALL KLOSE (ICHN,IER)
      CALL GCHN (ICHN,IER)  ;  GET RDOS CHANNEL
      CALL OPENA (ICHN,"INDEXY",0,IER)
      CALL WRS (ICHN,IEND,2,IER)  ;  ENDING FOR AFOS PRODUCT
      CALL KLOSE (ICHN,IER)
      CALL FSTORE ("INDEXY",0,IER)  ;  STORE INTO ERHWRKTPB
      CALL FORKP ("STAB","ERHWRKTPB",IER)  ;  TURN ON ALERT LIGHT
28    CONTINUE
      CALL DFILW ("INDEXX",IER)
      CALL DFILW ("INDEXY",IER)
      STOP
      END
```

*

*

```
      SUBROUTINE DECOS (Q)
C   DECODES RAOB SIGNIFICANT LEVELS UP TO 100MB
      COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
      DIMENSION IOUT(40)
      INTEGER Q
      K=0
      CALL AFREAD (1,JST,$100)
      CALL AFREAD (2,IOUT,$50,$125)
      IF (IOUT(6).EQ." U".AND.IOUT(7).EQ."J1") K=K+4
      K1=K/2
C   TEST FOR MISSING RAOB 10142
      IIA=INTCVT(30+K,4)
      IIB=INTCVT(34+K,1)
      IIIA=INTCVT(36+K,4)
      IF (IIA.EQ.5151.AND.IIB.EQ.5.AND.IIIA.EQ.1014) GO TO 126
      IF (IOUT(9+K1).EQ." 5".OR.IOUT(9+K1).EQ." 6".OR.IOUT(9+K1).EQ." 7".
     1 OR.IOUT(9+K1).EQ." 8") GO TO 8  ;  TESTING FOR DOUBLE SPACE AFTER TTBB
      K=K-1
8     JDATE=INTCVT(18+K,2)-50
      JHOUR=INTCVT(20+K,2)
      JNO=0
```

```
      DO 1 I=0,2
      I1=6*I
      I2=I1+I1
      II=INTCVT(30+K+I2,2)
      IJJ=I*11
      IF (IJJ.NE.II) GO TO 127  ;  TEST FOR IMPROPER FORMAT
      IF (IOUT(17+K1+I1).EQ."//".OR.IOUT(17+K1+I1).EQ."/ ") GO TO 1 ; SIG LVL PRS MISG
      P(JNO)=FLTCVT(32+K+I2,3)
      IF (P(JNO).LT.100.) P(JNO)=P(JNO)+1000.
      TS(JNO)=FLTCVT(36+K+I2,3)
      IF (IOUT(20+K1+I1).EQ."//".OR.IOUT(20+K1+I1).EQ."/ ") GO TO 2  ;  DEWPT MISG
      TSD(JNO)=FLTCVT(39+K+I2,2)
      GO TO 3
2     TSD(JNO)=999.
3     CALL TEMP1(TS(JNO),TSD(JNO))  ;  TEMPERATURE DECODER
      JNO=JNO+1
1     CONTINUE
C   FIRST LINE OF RAOB IS FINISHED HERE
C   STATEMENT 4 STARTS 2ND AND SUBSEQUENT LINES
      JNO=JNO-1
      IJK=22
4     CALL AFREAD (2,IOUT,$50,$125)
      DO 5 I=0,4
      I1=6*I
      I2=I1+I1
      IJK=IJK+11
      IF (IJK.EQ.110) IJK=11
      IJK1=I2+1
      IF (IOUT(I1+2).EQ."//") GO TO 5  ;  SIG LVL PRESSURE MISG
      I515=INTCVT(I2+1,3)
      IF (I515.EQ.515) GO TO 53  ;  TEST FOR 51515 101XX GROUP ENDING MSG
      JI=INTCVT(IJK1,2)
      IF (IJK.NE.JI) GO TO 128  ;  TEST FOR IMPROPER FORMAT
      IF (JNO.EQ.50) GO TO 51
      JNO=JNO+1
      P(JNO)=FLTCVT(I2+3,3)
      IF (P(JNO).LT.100.) P(JNO)=P(JNO)+1000.
      TS(JNO)=FLTCVT(I2+7,3)
      IO=IOUT(I1+6)
      IF (IO.EQ."/ ".OR.IO.EQ."/=") GO TO 6
      TSD(JNO)=FLTCVT(I2+10,2)
      GO TO 7
6     TSD(JNO)=999.
7     CALL TEMP1(TS(JNO),TSD(JNO))
      IF (IO.EQ."/=".OR.IO.EQ."0=".OR.IO.EQ."1=".OR.IO.EQ."2=".OR.IO.EQ.
     1 "3=".OR.IO.EQ."4=".OR.IO.EQ."5=".OR.IO.EQ."6=".OR.IO.EQ."7=".OR.IO.
     2 EQ."8=".OR.IO.EQ."9=") GO TO 53  ;  TEMPERATURES FINISHED
5     CONTINUE
      GO TO 4   ;   RETURNS TO 4 TO DO 3RD AND SUBSEQUENT LINES
53    CONTINUE
C   IF WIND DATA REQUIRED, READ IT HERE...
      RETURN
51    WRITE (10,52) P(JNO)
52    FORMAT (1H0,"51 SIGNIFICANT LEVELS HAVE BEEN DECODED, LEVELS ABOVE
     1 ",F5.0,"MB DISREGARDED.")
      RETURN
50    TYPE "AFREAD ERROR 50 - DECOS"
      RETURN Q
100   TYPE "AFREAD ERROR 100 - DECOS"
      RETURN Q
```

```
125     TYPE "AFREAD ERROR 125 - DECOS"
        RETURN Q
126     TYPE "STATION MISSING"
        RETURN Q
127     WRITE (10,129) IJJ,II
129     FORMAT (1H ,"UJ1 DATA NOT IN PROPER FORMAT, IJJ = ",I4," II = ",I4)
        RETURN Q
128     WRITE (10,130) IJK,JI
130     FORMAT (1H ,"UJ1 DATA NOT IN PROPER FORMAT, IJK = ",I4," JI = ",I4)
        RETURN Q
        END
                                *

                                *
        SUBROUTINE TEMP1 (T,TD)
C   COMPUTES + OR - TEMPERATURE, AND COMPUTES DEWPOINT
        TT=AMOD(T,2.)
        IF (TT.EQ.1.) T=-T
        T=T*.1
        IF (TD.EQ.999.) RETURN
        IF (TD.LE.50.) GO TO 1
        TD=T-(TD-50.)
        RETURN
1       TD=T-TD*.1
        RETURN
        END
                                *

                                *
C   SUBROUTINE -RANN2- COMPUTES ENERGY AREAS ON THERMODYNAMIC DIAGRAM,
C   USING PARCEL METHOD, WITH SELECTED ENTRAINMENT RATE AND PRESSURE STEP
C   JNOJ = NO. OF LEVELS IN RAOB: PA,TSA,TSDA
C   PX = PRESSURE LEVEL ENDING "B1" INDEX COMPUTATION
        SUBROUTINE RANN2 (PA,TSA,TSDA,JNOJ,PX)
        COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
        COMMON/G/PP(0:20),ET(20),TW(0:50),DP,EFF,KMOD,KK
        COMMON/T/RLCL,RLFC,EL,B2,B2P,B2N,IALL,B1,B1P,B1N,EX
        COMMON/CCL/PCCL,ETCCL,TS0,TSD0,L,TSCCL,TCCL,TDCCL,WAVG
        DIMENSION PA(0:50),TSA(0:50),TSDA(0:50)
        THETA(T,P2,P1)=T*(P2/P1)**.2857142 ; DRY ADIABATIC (T,P1) TO (THETA,P2)
        R=287.04  ;  GAS CONSTANT FOR DRY AIR .. J/KG PER DEG K
        R=R*.1    ;  SCALING ENERGY UNITS
        EF1=.00002*EFF  ;  ENTRAINMENT FACTOR PER MILLIBAR
        KMOD=KMOD-1   ;    KMOD= -1, WHEN OPERATING ON CCL MODIFIED RAOB
        IF (KMOD.EQ.-1) GO TO 106
        TS0=TSA(0)
        TSD0=TSDA(0)
106     IF (TS0.NE.TSD0) GO TO 92
        TC=TS0  ;  PARCEL INITIALLY SATURATED
        RLCL=PA(0)
        GO TO 107
92      TC=TCONOF(TS0,TSD0)  ;  CONDENSATION TEMP
        PC=PA(0)*((TC+273.16)/(TS0+273.16))**(1./.2857142) ; COND. PRES.
        IF (KMOD.EQ.-1) PC=PCCL ;  PC COMPUTED ABOVE IS NOT EXACTLY PCCL
107     TH=THETA(TS0+273.16,1000.,PA(0))-273.16  ;  POT. TMP DEG C
        WTH=WOBF(TH)
        WTC=WOBF(TC)
        THW=TH-WTH+WTC  ;  EQUIV WET BULB POT TMP (DEG C)
```

```
C    LIFT DRY ADIABATICALLY UNTIL TP=TC AT PRESSURE PC
          DO 7 I=1,20
7         ET(I)=0.
          DT1=0.
          IF (KMOD.EQ.-1) DT1=TS0-TSA(0)
          J=0
          JJ=0
          JK=0
          KJ=0
          EN=0.
          EP=0.
          P1=PA(J)
          PP(0)=PA(0)
          KK=1
          KKK=0
          TP=TSA(J)
          IF (KMOD.EQ.-1) TP=TS0
          IF (TSDA(J).EQ.999.) TSDA(J)=TSA(J)-30. ; IF MISG, ASSUME DRY
          WP=WMROF(P1,TSDA(J))
          IF (IALL.EQ.2) WRITE (10,86)
86        FORMAT (1H ,"P1",8X,"P2",10X,"TE",13X,"TP",13X,"DT1",12X,"DT2",
     1     12X,"E")
          IF (TS0.EQ.TSD0) GO TO 15  ;  PARCEL INITIALLY SATURATED
13        P2=P1-DP
          MJ=0
          IF (PC-P2) 3,4,4
4         P2=PC
          RLCL=PC  ;   LIFTING CONDENSATION LVL
3         IF (PA(J+1)-P2) 5,6,6
6         P2=PA(J+1)
          J=J+1
          MJ=1
          KJ=1
30        PLOG1=ALOG(PA(J)/PA(J+1))
          FACTORT=(TSA(J)-TSA(J+1))/PLOG1
          IF (TSDA(J+1).EQ.999.) TSDA(J+1)=TSA(J+1)-30. ; IF MISG, ASSUME DRY
          FACTORD=(TSDA(J)-TSDA(J+1))/PLOG1
          KJ=1
5         IF (KJ.EQ.0) GO TO 30  ;  INSURES FACTORT,D COMPUTED 1ST TIME THRU
          IF (JJ.EQ.0) TP0=TP        ;   SAVE ORIGINAL TP
          IF (JJ.EQ.1) TP=TP0  ; RESETS TP TO ORIGINAL VALUE, IF P2 ADJUSTED
          TP=TP+273.16   ;   CONVERT TO DEG K
          TP=THETA (TP,P2,P1)-273.16 ;DRY ADIABATIC LIFT P1 TO P2 DEG C
          PLOG2=ALOG(P2/PA(J+1))
          TE=TSA(J+1)+PLOG2*FACTORT ; ENVIRONMENTAL TEMP AT P2
          DP1=P1-P2
          IF (KMOD.EQ.-1) GO TO 42  ;  NO ENTRAINMENT BELOW CCL LEVEL
          IF (EFF.EQ.0.) GO TO 42  ;   EFF=0. FOR NO ENTRAINMENT
          EF=EF1*DP1
          TP=(TP+273.16+EF*(TE+273.16))/(1.+EF)-273.16  ;  DEG C
          TDE=TSDA(J+1)+PLOG2*FACTORD  ; DEG C
          WE=WMROF(P2,TDE)  ;  G/KG MIXING RATIO OF ENVIRONMENT
          WP=(WP+EF*WE)/(1.+EF)  ; MIXING RATIO OF PARCEL AFTER MIXING
          X=.0200*(TP-12.5+7500./P2)      ;  CORRECTION FOR NON-IDEAL GAS
          WFW=1.+.0000045*P2+.00140*X*X  ;  CORRECTION FOR NON-IDEAL GAS
          E2=WP*.001*P2/((WP*.001+.62197)*WFW) ; VAPOR PRES (MB) OF PARCEL
          ES2=VAPFW(TP)   ;  SATURATION VAPOR PRES OF PARCEL
          ES=ES2-E2
          IF (ES) 40,40,41  ;  GOES TO 40, IF PARCEL SATURATED
41        IF (ES.LE..01) GO TO 40  ;  CLOSE ENOUGH FOR SATURATION
```

```
      TDP=DPTOF(E2)    ;  DEWPOINT OF PARCEL AFT MXG (DEG C)
      TC=TCONOF (TP,TDP)
      PC=P2*((TC+273.16)/(TP+273.16))**(1./.2857142)
      GO TO 42
40    TC=TP
      PC=P2
      RLCL=PC
C   SINCE LCL HAS BEEN CHANGED, NEW -THW- IS ALSO REQUIRED
      TH=THETA(TC+273.16,1000.,PC)-273.16  ;  POT TEMP  DEG C
      WTH=WOBF(TH)
      WTC=WOBF(TC)
      THW=TH-WTH+WTC
42    DT2=TP-TE
      IF (JJ.EQ.0) GO TO 96
      TI=DT1-DT2
      JJ=0
      JK=1
      IF (TI) 10,10,11
96    IF (KMOD.EQ.0) GO TO 14
      IF (P2.GT.PC) GO TO 14
      KKK=1
      GO TO 11    ;   LAST STEP IN COMPUTING CCL ENERGY
14    IF (JK.NE.1) GO TO 66 ; JK=1, IF PREVIOUS PASS WAS A CROSSING PT
      JK=0
      IF (DT2) 10,10,11
66    IF (DT2) 8,8,9
8     IF (DT1) 10,10,12
9     IF (DT1) 12,11,11
C   GOES TO 12 IF DRY ADIABAT CROSSES ENVIRONMENTAL TEMP
12    P2=P1-ABS(DT1)/(ABS(DT1)+ABS(DT2))*DP1 ; APPROX PRES WHERE DT2=0.
      IF (KK.LE.20) GO TO 75
      TYPE "ET DIMENSION EXCEEDS 20"
      STOP 75
75    KKK=1
      JJ=1
      IF (MJ.EQ.0) GO TO 5
C   MJ=1 MEANS J, WHICH HAS JUST BEEN SET AT STATEMENT 6, MUST BE RESET
C    TO INTERPOLATE PROPERLY
      J=J-1
      MJ=0
      GO TO 30
10    E=.5*(DT2+DT1)*ALOG(P1/P2)
      EN=EN+E
      IF (IALL.EQ.2) WRITE (10,85) P1,P2,TE,TP,DT1,DT2,E
85    FORMAT (1H ,2F10.3,5E15.6)
      P1=P2
      DT1=DT2
      IF (KKK.EQ.0.AND.P2.NE.PA(JNOJ)) GO TO 62
      ET(KK)=EN*R   ;   CONVERTS TO J/KG UNITS
      PP(KK)=P2
      KK=KK+1
      EN=0.
      KKK=0
62    IF (P2.EQ.PC) GO TO 15  ;  PARCEL SATURATED
      GO TO 13
11    E=.5*(DT2+DT1)*ALOG(P1/P2)
      EP=EP+E
      IF (IALL.EQ.2) WRITE (10,85) P1,P2,TE,TP,DT1,DT2,E
      P1=P2
      DT1=DT2
```

```
      IF (KKK.EQ.0.AND.P2.NE.PA(JNOJ)) GO TO 63
      ET(KK)=EP*R    ;   CONVERTS TO J/KG UNITS
      PP(KK)=P2
      KK=KK+1
      EP=0.
      KKK=0
63    IF (P2.EQ.PC) GO TO 15  ;  PARCEL SATURATED
      GO TO 13
C
C   LIFT PARCEL ALONG SATURATION ADIABATIC
C
15    CONTINUE
      IF (KMOD.EQ.0) GO TO 84
      DT1=0.
      ETCCL=ET(1)
      RETURN    ;   REMOVE, IF FULL COMPUTATION OF CCL MODIFIED SOUNDING IS DESIRED
      KK=1  ;  KK SET FROM 2 BACK TO 1, CCL ENERGY HAS JUST BEEN COMPUTED
84    JJ=0
      JK=0
      ISTOP=0
      KKK=0
24    P2=P1-DP
      MJ=0
      IF (PA(J+1)-P2) 16,17,17
17    P2=PA(J+1)
      IF (PA(J+1).GT.PA(JNOJ)) GO TO 25
      ISTOP=1
      GO TO 16
25    J=J+1
      MJ=1
88    PLOG1=ALOG(PA(J)/PA(J+1))
      FACTORT=(TSA(J)-TSA(J+1))/PLOG1
      IF (TSDA(J+1).EQ.999.) TSDA(J+1)=TSA(J+1)-30. ; IF MISG, ASSUME DRY
      FACTORD=(TSDA(J)-TSDA(J+1))/PLOG1
      KJ=1
16    IF (KJ.EQ.0) GO TO 88
      IF (JJ.EQ.0) THW0=THW        ; SAVE ORIGINAL THW
      IF (JJ.EQ.1) THW=THW0 ;RESETS THW TO ORIGINAL VALUE, IF P2 ADJUSTED
      TP=SATLFT (THW,P2)  ;  TEMP OF PARCEL AT P2 ON -THW- WET ADIABAT
      PLOG2=ALOG(P2/PA(J+1))
      TE=TSA(J+1)+PLOG2*FACTORT  ;  ENVIRONMENTAL TEMP AT P2
      DP1=P1-P2
      IF (EFF.EQ.0.) GO TO 67  ;  EFF=0. FOR NO ENTRAINMENT
      TDE=TSDA(J+1)+PLOG2*FACTORD ; ENVIRONMENTAL DEWPT AT P2
      WE=WMROF (P2,TDE)  ;  MIXING RATIO (G/KG) OF ENVIRONMENT
      WP=WMROF (P2,TP)   ;  MIXING RATIO OF SATURATED PARCEL
      EF=EF1*DP1
      WP=(WP+EF*WE)/(1.+EF)
      TP=(TP+273.16+EF*(TE+273.16))/(1.+EF)-273.16
      X=.0200*(TP-12.5+7500./P2)      ;  CORRECTION FOR NON-IDEAL GAS
      WFW=1.+.0000045*P2+.00140*X*X  ;  CORRECTION FOR NON-IDEAL GAS
      E2=WP*.001*P2/((WP*.001+.62197)*WFW) ; VAPOR PRES (MB) OF PARCEL
      TDP=DPTOF(E2)  ;  DEWPT OF PARCEL AFT MXG
      IF (TDP.GT.TP) TDP=TP
      TC=TCONOF(TP,TDP)
      TH=THETA(TP+273.16,1000.,P2)-273.16 ;  POT TEMP DEG C
      WTH=WOBF(TH)
      WTC=WOBF(TC)
      THW=TH-WTH+WTC  ;  EQUIV WET BULB POT TEMP (DEG C)
      TP=SATLFT(THW,P2)  ;  PARCEL TEMP AFT EVAPORATING LIQUID WATER
```

```
67     DT2=TP-TE
C   IF ADDITIONAL INFORMATION ON LEVELS IS NEEDED,  INSERT PRINT STATEMENT HERE
       IF (JJ.EQ.0) GO TO 23
       TI=DT1-DT2
       JJ=0
       JK=1
       IF (TI) 20,20,22
23     IF (JK.NE.1) GO TO 65 ; JK=1, IF PREVIOUS PASS WAS A CROSSING PT
       JK=0
C
C   IN CASE SAT. ADIABAT INTERSECTS ENVIRONMENTAL TEMP IN 2 PLACES CREATING
C   A VERY SMALL POSITIVE AREA, THIS AREA WILL BE IGNORED.
       CHECK=DT2*ET(KK-1)   ;  USUALLY NEGATIVE
       IF (CHECK.LT.0.) GO TO 100
       IF (DT2) 101,101,102
101    EN=ET(KK-1)
       KK=KK-1
       GO TO 20
102    TYPE "ERROR IN RANN2 SUBROUTINE, STATEMENT 102"
       STOP 102
C
100    IF (DT2) 20,20,22
65     IF (DT2) 18,18,19
18     IF (DT1) 20,20,21
19     IF (DT1) 21,22,22
C   GOES TO 21 IF WET ADIABAT CROSSES ENVIRONMENTAL TEMP
21     P2=P1-ABS(DT1)/(ABS(DT1)+ABS(DT2))*DP1
       IF (KK.LE.20) GO TO 76
       TYPE "ET DIMENSION EXCEEDS 20"
       STOP 76
76     KKK=1
       JJ=1
       IF (MJ.EQ.0) GO TO 16
C   MJ=1 MEANS J, WHICH HAS JUST BEEN SET AT STATEMENT 25, MUST BE RESET
C     TO INTERPOLATE PROPERLY
       J=J-1
       MJ=0
       GO TO 88
20     E=.5*(DT2+DT1)*ALOG(P1/P2)
       EN=EN+E
       IF (IALL.EQ.2) WRITE (10,85) P1,P2,TE,TP,DT1,DT2,E
       P1=P2
       DT1=DT2
       IF (P2.NE.PX) GO TO 99
       EX=EN*R  ;  SUBTOTAL FOR ENERGY AREA ENDING AT PX
       KX=KK
99     IF (KKK.EQ.0.AND.P2.NE.PA(JNOJ)) GO TO 60
       ET(KK)=EN*R   ;   CONVERTING TO J/KG UNITS
       PP(KK)=P2
       KK=KK+1
       EN=0.
       KKK=0
60     IF (ISTOP.EQ.1.AND.P1.EQ.PA(JNOJ)) GO TO 26
       GO TO 24
22     E=.5*(DT2+DT1)*ALOG(P1/P2)
       EP=EP+E
       IF (IALL.EQ.2) WRITE (10,85) P1,P2,TE,TP,DT1,DT2,E
       P1=P2
       DT1=DT2
       IF (P2.NE.PX) GO TO 104
```

```
      EX=EP*R  ;  SUBTOTAL FOR ENERGY AREA ENDING AT PX
      KX=KK
104   IF (KKK.EQ.0.AND.P2.NE.PA(JNOJ)) GO TO 61
      ET(KK)=EP*R   ;   CONVERTING TO J/KG UNITS
      PP(KK)=P2
      KK=KK+1
      EP=0.
      KKK=0
61    IF (ISTOP.EQ.1.AND.P1.EQ.PA(JNOJ)) GO TO 26
      GO TO 24
26    CONTINUE
C
C   KK = NUMBER OF ENERGY AREAS IN SOUNDING + 1
      KK1=KK-1
      KK2=KK-2
      KK3=KK-3
      EL=0.
      B2=99999.
      B2P=99999.
      B2N=99999.  ;   99999 DENOTES THAT VARIABLE IS UNDEFINED
      RLFC=0.
C   DETERMINE -LFC- LEVEL
      IF (KK.EQ.2.AND.ET(1).GT.0.) RLFC=PP(0)
      IF (KK.EQ.3.AND.ET(1).LT.0.) RLFC=PP(1)
      IF (KK.GE.4.AND.ET(1).LT.0.) RLFC=PP(1)
      IF (KK.GE.4.AND.ET(1).GT.0.) RLFC=PP(2)
C   IN ALL OTHER CASES RLFC IS UNDEFINED...RLFC=0.
C
      IF (ET(KK1).GT.0.) GO TO 70  ;  HIGHEST AREA IS +, NO INDICES COMPUTED
C   -EL- LEVEL DETERMINED HERE
      EL=PP(KK2)
C   COMPUTE ENERGY INDICES BELOW EL LEVEL
      IF (KK.EQ.2) GO TO 70  ;  ONLY ONE LAYER, ALL NEGATIVE
      B2=0.
      B2P=0.
      B2N=0.
      IF (ET(1).LT.0.) GO TO 58
      DO 74 I=1,KK2,2
74    B2P=B2P+ET(I)
      IF (KK.GT.3) GO TO 68
      B2N=0.
      GO TO 69
68    DO 73 I=2,KK3,2
73    B2N=B2N+ET(I)
      GO TO 69
58    DO 91 I=1,KK3,2
91    B2N=B2N+ET(I)
      DO 103 I=2,KK2,2
103   B2P=B2P+ET(I)
69    B2=B2P+B2N
70    CONTINUE
C
C   COMPUTE B1 INDEX (ENERGY AREAS ENDING AT PX)
      KX1=KX-1
      B1=0.
      B1P=0.
      B1N=0.
      DO 105 I=1,KX1
      IF (ET(I).LT.0.) B1N=B1N+ET(I)
      IF (ET(I).GT.0.) B1P=B1P+ET(I)
```

```
105    CONTINUE
       IF (EX.LT.0.) B1N=B1N+EX
       IF (EX.GT.0.) B1P=B1P+EX
       B1=B1P+B1N
       RETURN
       END
```

*

*

```
       SUBROUTINE BNDX (Q)
C   DETERMINES LEVEL OF MAXIMUM INSTABILITY IN LOWER 150MBS OF RAOB,
C   ADJUSTS ORIGINAL RAOB, SO LEVEL OF MAX INSTABILITY IS FIRST SGFNT
C   LEVEL AND ADDS ADDITIONAL PRES LEVEL PX, IF PX IS NOT A SGFNT LEVEL.
C   IF RAOB TERMINATES BELOW PX, IT IS EXTRAPOLATED TO PX, IF TOP LEVEL IS WITHIN 50 MBS
       COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
       COMMON/TT/PT(0:50),TST(0:50),TSDT(0:50)
       COMMON/V/JNOM,PX
       DIMENSION PB(2),TB(2),TDB(2)
       INTEGER Q
       THETA(T,P2,P1)=T*(P2/P1)**.2857142  ; DRY ADIABATIĆ (T,P1) TO (THETA,P2)
       DP2=150.
C   GET TEMP AND DEWPT AT P(0)-DP2 AND PX
       PB(1)=P(0)-DP2
       PB(2)=PX
       DO 5 J=1,2
       DO 4 I=0,JNO
       IF (PB(J)-P(I)) 4,6,7
4      CONTINUE
       I=I-1
       IF (J.EQ.2.AND.(P(JNO)-PB(2)).LT.50.) GO TO 7 ; EXTRAPOLATES, IF WITHIN 50MBS
       TYPE "RAOB TERMINATES TOO SOON. P(JNO) = ",P(JNO)
       RETURN Q
6      TB(J)=TS(I)
       TDB(J)=TSD(I)
       GO TO 5
7      FACTOR=ALOG(PB(J)/P(I))/ALOG(P(I-1)/P(I))
       TB(J)=TS(I)+FACTOR*(TS(I-1)-TS(I))
       TDB(J)=TSD(I)+FACTOR*(TSD(I-1)-TSD(I))
5      CONTINUE
C   FIND LARGEST POTENTIAL WET BULB TEMPERATURE IN FIRST DP2 MBS
       THWMAX=-1000.
       II=0
       DO 1 I=0,JNO
       IF (P(I)-PB(1)) 8,10,10
10     TC=TCONOF (TS(I),TSD(I))
       TH=THETA(TS(I)+273.16,1000.,P(I))-273.16 ; DEG C
       WTH=WOBF(TH)
       WTC=WOBF(TC)
       THW=TH-WTH+WTC  ;  WET BULB POTENTIAL TEMPERATURE
       IF (THW-THWMAX) 1,1,2
2      THWMAX=THW
       II=I
       PMAX=P(I)
       TMAX=TS(I)
       TDMAX=TSD(I)
1      CONTINUE
8      IF (P(I-1).EQ.PB(1)) GO TO 9
       TC=TCONOF (TB(1),TDB(1))
       TH=THETA(TB(1)+273.16,1000.,PB(1))-273.16  ;  DEG C
```

```
      WTH=WOBF(TH)
      WTC=WOBF(TC)
      THW=TH-WTH+WTC  ;  WET BULB POT TEMP
      IF (THW-THWMAX) 9,9,12
12    THWMAX=THW
      II=I-1
      PMAX=PB(1)
      TMAX=TB(1)
      TDMAX=TDB(1)
9     CONTINUE
C   MODIFY RAOB SO LOWEST LEVEL HAS MAXIMUM WET BULB POTENTIAL TEMPERATURE
      PT(0)=PMAX
      TST(0)=TMAX
      TSDT(0)=TDMAX
      JNOO=JNO-II
      DO 11 J=1,JNOO
      PT(J)=P(J+II)
      TST(J)=TS(J+II)
      TSDT(J)=TSD(J+II)
11    CONTINUE
      DO 14 J=1,JNOO
      IF (PB(2)-PT(J)) 14,17,16
14    CONTINUE
      J=JNOO+1
      GO TO 20  ;  EXTRAPOLATE RAOB
16    I=JNOO
18    TST(I+1)=TST(I)  ;  MOVE ALL LEVELS ABOVE PB(2) UP 1 LEVEL
      TSDT(I+1)=TSDT(I)
      PT(I+1)=PT(I)
      I=I-1
      IF (I.GE.J) GO TO 18  ;  J SET IN DO 14 LOOP
20    TST(J)=TB(2)  ;   ADD TB(2) LEVEL
      TSDT(J)=TDB(2)
      PT(J)=PB(2)
      JNOM=JNOO+1
      GO TO 19
17    JNOM=JNOO
19    CONTINUE
      RETURN
      END
```

*

*

```
      SUBROUTINE INDX1 (RLI,RKI,RWI,Q)
C   COMPUTES LIFTED INDEX, K INDEX, AND SHOWALTER INDEX
      COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
      DIMENSION PL(3),TL(3),TDL(3)
      INTEGER Q
      THETA(T,P2,P1)=T*(P2/P1)**.2857142 ; DRY ADIABATIC (T,P1) TO (THETA,P2)
      PL(1)=850.
      PL(2)=700.
      PL(3)=500.
      DP1=50.    ;   AVERAGES OVER FIRST -DP1- MBS.
      WSUM=0.
      THSUM=0.
      J=0
      P1=P(0)
      TE1=TS(0)
      TDE1=TSD(0)
```

```
      TH1=THETA(TE1+273.16,1000.,P(0))    ;   POT TEMP
      W1=WMROF(P(0),TDE1)
      PFINISH=P1-DP1
3     P2=PFINISH
      IF (P(J+1)-P2) 1,2,2
2     P2=P(J+1)
      J=J+1
1     PLOG1=ALOG(P(J)/P(J+1))
      FACTORT=(TS(J)-TS(J+1))/PLOG1
      FACTORD=(TSD(J)-TSD(J+1))/PLOG1
      PLOG2=ALOG(P2/P(J+1))
      TE2=TS(J+1)+FACTORT*PLOG2   ;  ENVIRONMENT TEMP AT P2
      TDE2=TSD(J+1)+FACTORD*PLOG2 ;  ENVIRONMENT DEWPT AT P2
      TH2=THETA(TE2+273.16,1000.,P2) ;  POT TEMP AT TE2,P2
      W2=WMROF(P2,TDE2)              ;  MIXING RATIO AT P2
      PLOG3=ALOG(P1/P2)
      TH=.5*(TH1+TH2)*PLOG3          ;  AVG POT TEMP IN LYR P1-P2
      W=.5*(W1+W2)*PLOG3             ;  AVG MIX RATIO IN LYR P1-P2
      THSUM=THSUM+TH
      WSUM=WSUM+W
      P1=P2
      TH1=TH2
      W1=W2
      IF (P2.GT.PFINISH) GO TO 3
C   COMPUTE AVG VALUES FOR FIRST -DP1- MBS.
      PLOG4=ALOG(P(0)/PFINISH)
      THAVG=THSUM/PLOG4
      WAVG=WSUM/PLOG4
      PPARCEL=P(0)-.5*DP1
      TPARCEL=THETA(THAVG,PPARCEL,1000.)-273.16  ;  DEG C
      X=.0200*(TPARCEL-12.5+7500./PPARCEL) ;  NON-IDEAL GAS CORRECTION
      WFW=1.+.0000045*PPARCEL+.00140*X*X   ;  NON-IDEAL GAS CORRECTION
      E2=.001*WAVG*PPARCEL/((WAVG*.001+.62197)*WFW)  ;  VAPOR PRES (MB)
      TDPARCEL=DPTOF(E2)
      TC=TCONOF(TPARCEL,TDPARCEL)
      TH=THAVG-273.16  ;  POT TEMP DEG  C
      WTH=WOBF(TH)
      WTC=WOBF(TC)
      THW=TH-WTH+WTC  ;  EQUIV WET BULB POT TEMP (DEG C)
      TP500=SATLFT(THW,500.)
C   GET TEMP AND DEWPT AT 850,700,500 MBS
      DO 5 J=1,3
      DO 4 I=0,JNO
      IF (PL(J)-P(I)) 4,6,7
4     CONTINUE
      TYPE "RAOB TERMINATES TOO SOON, P(JNO) = ",P(JNO)
      RETURN Q
6     TL(J)=TS(I)
      TDL(J)=TSD(I)
      GO TO 5
7     FACTOR=ALOG(PL(J)/P(I))/ALOG(P(I-1)/P(I))
      TL(J)=TS(I)+FACTOR*(TS(I-1)-TS(I))
      TDL(J)=TSD(I)+FACTOR*(TSD(I-1)-TSD(I))
5     CONTINUE
      RLI=TL(3)-TP500                          ;  LIFTED INDEX
      RKI=(TL(1)-TL(3))+TDL(1)-(TL(2)-TDL(2))  ;  K INDEX
C   COMPUTE SHOWALTER INDEX
      TC=TCONOF(TL(1),TDL(1))
      TH=THETA(TL(1)+273.16,1000.,850.)-273.16  ;  DEG C
      WTH=WOBF(TH)
```

```
      WTC=WOBF(TC)
      THW=TH-WTH+WTC  ;  EQUIV WET BULB POT TEMP
      TP=SATLFT(THW,500.)
      RWI=TL(3)-TP     ;   SHOWALTER INDEX
      RETURN
      END
```

*

*

```
      SUBROUTINE CCL1
C   COMPUTES CCL AND CONVECTIVE TEMPERATURE
      COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
      COMMON/CCL/PCCL,ETCCL,TS0,TSD0,L,TSCCL,TCCL,TDCCL,WAVG
      THETA(T,P2,P1)=T*(P2/P1)**.2857142 ; DRY ADIABATIC (T,P1) TO (THETA,P2)
      DP1=100.    ;   AVERAGES MIXING RATIO OVER FIRST -DP1- MBS.
      WSUM=0.
      J=0
      P1=P(0)
      TDE1=TSD(0)
      W1=WMROF(P(0),TDE1)
      PFINISH=P1-DP1
3     P2=PFINISH
      IF (P(J+1)-P2) 1,2,2
2     P2=P(J+1)
      J=J+1
1     PLOG1=ALOG(P(J)/P(J+1))
      FACTORD=(TSD(J)-TSD(J+1))/PLOG1
      PLOG2=ALOG(P2/P(J+1))
      TDE2=TSD(J+1)+FACTORD*PLOG2 ;  ENVIRONMENT DEWPT AT P2
      W2=WMROF(P2,TDE2)              ;  MIXING RATIO AT P2
      PLOG3=ALOG(P1/P2)
      W=.5*(W1+W2)*PLOG3             ;  AVG MIX RATIO IN LYR P1-P2
      WSUM=WSUM+W
      P1=P2
      W1=W2
      IF (P2.GT.PFINISH) GO TO 3
C   COMPUTE AVG VALUES FOR FIRST -DP1- MBS.
      PLOG4=ALOG(P(0)/PFINISH)
      WAVG=WSUM/PLOG4
      J=0
      WS1=WMROF(P(0),TS(0))
27    IF (WS1-WAVG) 20,20,21  ;  MAY GO TO 20, IF SFC INVERSION EXISTS
20    J=J+1
      WS1=WMROF(P(J),TS(J))
      GO TO 27
21    CONTINUE
      J=J+1
      WS2=WMROF(P(J),TS(J))
      IF (WS2-WAVG) 22,23,25
25    WS1=WS2
      GO TO 21
23    PCCL=P(J)
      TCCL=TS(J)
      GO TO 24
22    CONTINUE
C   MXG RATIO INTERSECTS ENVIRONMENTAL TEMP BTWN P(J) AND P(J-1)
C   THIS LAYER WILL BE SUBDIVIDED UNTIL SATURATION VAPOR PRESSURE AT
C   MIDPOINT OF LAYER IS SUFFICIENTLY CLOSE (.01 G/KG) TO WAVG.
C   THIS DETERMINES THE CCL LEVEL.
```

```
      P1=P(J-1)    ;    BOTTOM
      P2=P(J)      ;    TOP
      T1=TS(J-1)   ;    BOTTOM
      T2=TS(J)     ;    TOP
31    ALOG1=ALOG(P1/P2)
      PM=.5*(P1+P2)    ;    MIDPOINT PRESSURE
      ALOG2=ALOG(PM/P2)
      TPM=T2+(T1-T2)/ALOG1*ALOG2   ;  MIDPOINT TEMPERATURE
      WSM=WMROF(PM,TPM)    ;    MIDPOINT SATURATION MIXING RATIO
      IF(ABS(WSM-WAVG).LE..01)  GO TO 29    ;    TEST FOR TOLERANCE
      IF (WSM-WAVG) 28,29,30
28    P2=PM
      T2=TPM
      GO TO 31
30    P1=PM
      T1=TPM
      GO TO 31
29    PCCL=PM       ;    CCL PRESSURE
      TCCL=TPM      ;    CCL TEMPERATURE
24    CONTINUE
C   COMPUTE DEWPOINT AT CCL LEVEL
      ALOG1=ALOG(P(J-1)/P(J))
      ALOG2=ALOG(PM/P(J))
      TDCCL=TSD(J)+(TSD(J-1)-TSD(J))/ALOG1*ALOG2
      IF (TDCCL.GT.TCCL) TDCCL=TCCL   ;  CORRECTION FOR DEWPOINT EXCEEDING
     1 TEMPERATURE BY SMALL AMT
      TSCCL=THETA(TCCL+273.16,P(0),PCCL)-273.16   ;  CONVECTIVE TEMP DEG C
      L=J                    INDEX NUMBER OF ADDED CCL LEVEL
      RETURN
      END
```

*

*

```
C   THIS SUBROUTINE TO BE CALLED AFTER -CCL1- IS CALLED
      SUBROUTINE MODRB
      COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
      COMMON/CCL/PCCL,ETCCL,TS0,TSD0,L,TSCCL,TCCL,TDCCL,WAVG
C   MODIFY ORIGINAL RAOB FOR SOLAR HEATING BLO CCL
      TS0=TSCCL                    ;   SFC TEMP RESET
C   MOVE SGFNT LVLS ABV PCCL UP ONE LVL
      I=JNO
30    TS(I+1)=TS(I)
      TSD(I+1)=TSD(I)
      P(I+1)=P(I)
      I=I-1
      IF (I.GE.L) GO TO 30
C   ONE ADDITIONAL LVL ADDED AT PCCL
      TS(L)=TCCL
      TSD(L)=TDCCL
      P(L)=PCCL
C   MODIFY TSD(0) TO CONFORM TO WAVG, AVG MIXING RATIO IN LOWEST 100 MBS
      X=.0200*(TSD(0)-12.5+7500./P(0))   ; NON-IDEAL GAS CORRECTION
      WFW=1.+.0000045*P(0)+.00140*X*X    ; NON-IDEAL GAS CORRECTION
      E2=.001*WAVG*P(0)/((WAVG*.001+.62197)*WFW)   ;   VAPOR PRESSURE
      TSD0=DPTOF(E2)
      JNO=JNO+1
      RETURN
      END
```

*

*

```
      SUBROUTINE PULYR
C   COMPUTATION OF POTENTIAL (CONVECTIVE) UNSTABLE LAYERS, WITH LAPSE RATE
C   OF WET BULB POTENTIAL TEMPERATURE AND AMOUNT OF LIFT REQUIRED FOR SATURATION
      COMMON/S/JST(5),JDATE,JHOUR,JNO,JJNO,P(0:50),TS(0:50),TSD(0:50)
      COMMON/G/PP(0:20),ET(20),TW(0:50),DP,EFF,KMOD,KK
      COMMON/GG/NJ,PPB(15),PPT(15),DELPP(15),DTWDP(15),DPB(15),DPT(15),
     1 PTMAX,PBMAX,TWLAPSE,DMAX
      DMAX=0
      NJ=0
      IT=-1
      MK=0
      JNNO=JNO-1
      DO 1 I=0,JNNO
      IF ((TW(I)-TW(I+1)).LE.0.) GO TO 2  ;  GOES TO 2, IF STABLE
      GO TO 4
2     IF (MK.EQ.0) GO TO 1  ;  MK=0 INITIALLY, OR IF PREVIOUS LYR STABLE
      GO TO 3  ;  GOES TO 3, WHEN TOP OF UNSTABLE LYRS IS REACHED
C   DETERMINING INDICES OF UNSTABLE LYR, IT = TOP, IB = BOTTOM
4     IF (I.GT.IT) IB=I
      IT=I+1
      MK=1
      GO TO 1
3     NJ=NJ+1
      PPT(NJ)=P(IT)
      PPB(NJ)=P(IB)
      DELPP(NJ)=P(IB)-P(IT)
      DTWDP(NJ)=(TW(IB)-TW(IT))/DELPP(NJ)*100.
      IF(DELPP(NJ).LE.DMAX) GO TO 5
      DMAX=DELPP(NJ)
      TWLAPSE=DTWDP(NJ)
      PTMAX=PPT(NJ)
      PBMAX=PPB(NJ)
5     TC=TCONOF (TS(IB),TSD(IB))  ;  CONDENSATION TEMPERATURE
      PC=P(IB)*((TC+273.16)/(TS(IB)+273.16))**(1./.2857142) ; COND PRESSURE
      DPB(NJ)=P(IB)-PC  ;  AMT OF LIFT REQUIRED FOR BOTTOM SATURATION
      TC=TCONOF (TS(IT),TSD(IT))
      PC=P(IT)*((TC+273.16)/(TS(IT)+273.16))**(1./.2857142)
      DPT(NJ)=P(IT)-PC  ;  AMT OF LIFT REQUIRED FOR TOP SATURATION
      MK=0
1     CONTINUE
      RETURN
      END
```

*

*

```
      FUNCTION WOBF(T)
COMPUTE BY DOUBLE ASYMPTOTIC APPROXIMATION
CONSIDER SEPARATELY IF .GT. OR .LE. 20 DEG.
CENT. FOR ALL TEMPS...THETW=THETA-WOBF(THETA)+WOBF(TEMPCON)
CENT. FOR ALL TEMPS...THETM=THETA-WOBF(THETA)+WOBF(TEMP)
      X=T-20.0
      IF(X) 10,10,20
10    CONTINUE
CURVE FIG FOR COOL TEMPERATURE RANGE
      POL=1.000+X*(-8.8416605E-3+X*(1.4714143E-4+X*(-9.6719890E-7
     1 +X*(-3.2607217E-8+X*(-3.8598073E-10)))))
```

```
      POL=POL*POL
      WOBF=15.130/(POL*POL)
      RETURN
20    CONTINUE
CURVE FIT FOR WARMER TEMPERATURES
      POL=1.000+X*(3.6182989E-3+X*(-1.3603273E-5+X*(4.9618922E-7
     1 +X*(-6.1059365E-9+X*(3.9401551E-11+X*(-1.2588129E-13
     2 +X*(1.6688280E-16)))))))
      POL=POL*POL
      WOBF=29.930/(POL*POL)+0.9600*X-14.800
      RETURN
      END
                                   *


                                   *
      FUNCTION SATLFT (THM,P)
COMPUTES TEMPERATURE (DEG C) WHERE THETA MOIST (DEG C) CROSSES P (MB)
CONSIDER THE EXPONENTIAL FOR POTENTIAL TEMPERATURE AS ROCP
      ROCP=0.28571428
      IF(ABS(P-1000.0)-0.0010) 100,100,200
100   SATLFT=THM
      RETURN
200   PWRP=(P/1000.0)**ROCP
COMPUTE TEMPERATURE OF DRY ADIABATIC LIFT FOR FIRST GUESS
      TONE=(THM+273.16)*PWRP-273.16
CONSIDER PSEUDO-ADIABAT, EW1, THROUGH TONE AT P.
COMPUTE EONE=EW1-THM
      EONE=WOBF(TONE)-WOBF(THM)
      RATE=1.0
      GO TO 330
300   CONTINUE
CONTRIBUTION TO ITERATION IS CHANGE IN T
CORRESPONDING TO CHANGE IN E
      RATE=(TTWO-TONE)/(ETWO-EONE)
      TONE=TTWO
      EONE=ETWO
330   CONTINUE
COMPUTE ESTIMATED SATLIFT, TTWO
      TTWO=TONE-EONE*RATE
CONSIDER PSEUDO-ADIABAT, EW2, THROUGH TTWO AT P.
COMPUTE ETWO=EW2-THM
      ETWO=(TTWO+273.16)/PWRP-273.16
      ETWO=ETWO+WOBF(TTWO)-WOBF(ETWO)-THM
CORRECTION TO TTWO IS EOR
      EOR=ETWO*RATE
      IF(ABS(EOR)-0.1000) 400,400,300
400   SATLFT=TTWO-EOR
      RETURN
      END
                                   *


                                   *
      FUNCTION TCONOF(TEMP,DEWPT)
COMPUTES CONDENSATION TEMPERATURE (DEGREES CENT) BY LIFTING
      S=TEMP-DEWPT
CONSIDER TEMP AND DEWPT TO BE LIKE UNITS (C OR K)
      T=TEMP
      IF(100.-TEMP) 4,5,5
```

```
4       T=TEMP-273.16
COMPUTE CURVE FIT IN MOST EFFICIENT MANNER
5       DLT=S*(1.2185+0.001278*T+S*(-0.002190+11.73E-6*S-5.20E-6*T))
        TCONOF=T-DLT
        RETURN
        END
```

*

*

```
        FUNCTION WMROF(P,TD)
COMPUTE MIXING RATIO (G/KG)...DEWPOINT (DEGREES C OR K)...PRESSURE (MB)
        T=TD
        IF (100.-T) 3,4,4
3       T=T-273.16
CURVE FIT CORRECTION FOR NON-IDEAL GAS
4       X=0.0200*(T-12.5+7500.0/P)
        WFW=1.+0.0000045*P+0.00140*X*X
COMPUTE ACCORDING TO STANDARD FORMULA
        FWESW=WFW*VAPFW(T)
        WMROF=621.97*(FWESW/(P-FWESW))
        RETURN
        END
```

*

*

```
        FUNCTION DPTOF(EW)
COMPUTE DEWPOINT, DPT, IN DEGREES C GIVEN WATER VAPOR PRESSURE (MB)
CREATE TOLERANCE TO DEGREE DESIRED
        TOL=0.00010
        IF (EW-0.21382876E-09) 20,20,30
20      DPTOF=-10000.
        RETURN
30      IF (1013.0-EW) 20,100,100
CREATE GUESS BY INVERTING TETEN-S FORMULA
100     X=ALOG(EW/6.1078)
        BOT=17.269388-X
        DPTOF=(237.3*X)/BOT
        BOT=BOT*EW
        DELTM=0.
200     EDP=VAPFW(DPTOF)
CORRECT GUESS BY DERIVATIVE OF TEMPERATURE WITH RESPECT TO VAPOR PRES.
CALCULATED FROM INVERSE OF TETEN-S FORMULA
        DTDE=(DPTOF+237.3)/BOT
        DELT=DTDE*(EW-EDP)
        DPTOF=DPTOF+DELT
CHECK THAT ITERATION IS NOT IN AN ENDLESS CYCLE, A RARE SITUATION
C    IF NEEDED, CHANGE -TOL- AND EXIT
        DM=DELT-DELTM
        IF(ABS(DM).GE.1.E-7) GO TO 10 ; IF DM VERY SMALL, ITERATION IS ENDLESS
        TOL=ABS(DELT)
        TYPE "TOLERANCE (TOL) IN DPTOF CHANGED TO ",TOL," (NORMAL TOL = .00010)"
10      DELTM=-DELT
CHECK TO SEE IF ANSWER CLOSE ENOUGH, IF NOT ITERATE OVER CORRECTION
        IF (ABS(DELT)-TOL) 300,300,200
CHANGE SO DEWPOINT IS ALWAYS LESS THAN THE TEMP.
COMPATIBILITY WITH TOL IS FORCED
300     DPTOF=DPTOF-TOL
        RETURN
```

```
      END
                                  *

                                  *
      FUNCTION VAPFW(T)
COMPUTE SATURATION VAPOR PRESSURE OVER WATER, VAPFW, IN MBS.
CONSIDER T(TEMPERATURE) IN DEGREES C OR DEGREES K.
      X=T
      IF (100.0-X) 3,4,4
3     X=X-273.16
CURVE FIT FOR RANGE -50 < T < 100 DEGREES C.
4     POL = 0.99999683  E-00 + X *(-0.90826951  E-02 +
     1    X *(0.78736169  E-04 + X *(-0.61117958  E-06 +
     2    X *(0.43884187  E-08 + X *(-0.29883885  E-10 +
     3    X *(0.21874425  E-12 + X *(-0.17892321  E-14 +
     4    X *(0.11112018  E-16 + X *(-0.30994571  E-19))))))))))
      POL=POL*POL
      POL=POL*POL
      VAPFW=6.107800/(POL*POL)
      RETURN
      END
```

NOAA SCIENTIFIC AND TECHNICAL PUBLICATIONS

The National Oceanic and Atmospheric Administration was established as part of the Department of Commerce on October 3, 1970. The mission responsibilities of NOAA are to assess the socioeconomic impact of natural and technological changes in the environment and to monitor and predict the state of the solid Earth, the oceans and their living resources, the atmosphere, and the space environment of the Earth.

The major components of NOAA regularly produce various types of scientific and technical information in the following kinds of publications:

PROFESSIONAL PAPERS — Important definitive research results, major techniques, and special investigations.

CONTRACT AND GRANT REPORTS — Reports prepared by contractors or grantees under NOAA sponsorship.

ATLAS — Presentation of analyzed data generally in the form of maps showing distribution of rainfall, chemical and physical conditions of oceans and atmosphere, distribution of fishes and marine mammals, ionospheric conditions, etc.

TECHNICAL SERVICE PUBLICATIONS — Reports containing data, observations, instructions, etc. A partial listing includes data serials; prediction and outlook periodicals; technical manuals, training papers, planning reports, and information serials; and miscellaneous technical publications.

TECHNICAL REPORTS — Journal quality with extensive details, mathematical developments, or data listings.

TECHNICAL MEMORANDUMS — Reports of preliminary, partial, or negative research or technology results, interim instructions, and the like.

Information on availability of NOAA publications can be obtained from:

ENVIRONMENTAL SCIENCE INFORMATION CENTER (D822)
ENVIRONMENTAL DATA AND INFORMATION SERVICE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
U.S. DEPARTMENT OF COMMERCE

6009 Executive Boulevard
Rockville, MD 20852